Problem:
Score : 1100 points

Problem Statement

There are N strings of lowercase alphabet only. The i-th string is S_i. Every string is unique.

Provide answers for the Q queries below. The i-th query has the following format:

Query: An integer k_i and a string p_{i,1}p_{i,2}...p_{i,26} that results from permuting {a,b,...,z} are given. Output the sequence of the string S_{k_i} among the N strings in lexicographical order when the literal sequence is p_{i,1}<p_{i,2}<...<p_{i,26}.

Constraints

1 ≦ N,Q ≦ 100000

1 ≦ |S_i| (1 ≦ i ≦ N)

S_i (1 ≦ i ≦ N) is a string of lowercase alphabet.

The sum of |S_i| is no more than 400000.

Every S_i is unique.

1 ≦ k_i ≦ N (1 ≦ i ≦ Q)

For all 1 ≦ i ≦ Q, p_{i,1}p_{i,2}...p_{i,26} is a permutation of abcd...z.

Input

Inputs are provided from standard inputs in the following form.

N
S_1
:
S_N
Q
k_1 p_{1,1}p_{1,2}...p_{1,26}
:
k_Q p_{Q,1}p_{Q,2}...p_{Q,26}

Output

Output Q lines.

On line i, for the i-th query, output an integer indicating the sequence of the string S_{k_i} among the N strings in lexicographical order.

Sample Input 1

5
aa
abbaa
abbba
aaab
aaaaaba
5
1 abcdefghijklmnopqrstuvwxyz
2 bacdefghijklmnopqrstuvwxyz
3 abcdefghijklmnopqrstuvwxyz
4 bacdefghijklmnopqrstuvwxyz
5 abcdefghijklmnopqrstuvwxyz

Sample Output 1

1
2
5
4
2

When the literal sequence is a < b, sorting the input strings in lexicographical order yields aa, aaaaaba, aaab, abbaa, abbba. The answers to queries 1, 3, and 5 are thus 1, 5, and 2, respectively.

When the literal sequence is b < a, sorting the input strings in lexicographical order yields abbba, abbaa, aa, aaab, aaaaaba. The answers to queries 2 and 4 are thus 2 and 4, respectively.

Sample Input 2

8
abrakatabra
abadaba
abracadabra
atcoder
grand
contest
ababa
a
6
3 abcdefghijklmnopqrstuvwxyz
6 qwertyuiopasdfghjklzxcvbnm
8 poiuytrewqlkjhgfdsamnbvcxz
2 qazwsxedcrfvtgbyhnujmikolp
1 plokmijnuhbygvtfcrdxeszwaq
4 mnbvcxzasdfghjklpoiuytrewq

Sample Output 2

4
8
2
3
4
7

Solution in Pascal (FPC 2.6.2) (judge verdict: Accepted)

//{$R-,Q-,S-,I-}
{$OPTIMIZATION LEVEL2}
{$INLINE ON}

uses math;

procedure mofile;
begin
        {$IFDEF ONLINE_JUDGE}
        assign(input,'');
        reset(input);
        assign(output,'');
        rewrite(output);
        {$ELSE}
        assign(input,'checking.inp');
        reset(input);
        assign(output,'');
        rewrite(output);
        {$ENDIF}
end;

procedure dongfile;
begin
        close(input);
        close(output);
end;

const
        maxc=1000000000+69;
        modulo=1000000000+7;
        maxn=400000+10;

type
        re=record
                x,y:longint;
        end;

var
        n,q,dem,res:longint;
        s:array[0..maxn] of ansistring;
        a,c:array[0..maxn*4,'`'..'z'] of longint;
        cnt:array[0..maxn,'`'..'z','`'..'z'] of longint;
        al:ansistring;

procedure insert(p,x,i:longint);
begin
        if x=length(s[i])+1 then exit;
        inc(c[p][s[i,x]]);
        if a[p][s[i,x]]>0 then insert(a[p][s[i,x]],x+1,i) else
        begin
                inc(dem);
                a[p][s[i,x]]:=dem;
                insert(dem,x+1,i);
        end;
end;

procedure nhapdl;
var
        i:longint;
begin
        readln(n);
        for i:=1 to n do
        begin
                readln(s[i]);
                s[i]:=s[i]+'`';
                insert(0,1,i);
        end;
end;

procedure dfs(p,x,i:longint);
var
        j:char;
begin
        if x=length(s[i])+1 then exit;
        for j:='`' to 'z' do
        begin
                inc(cnt[i,j,s[i,x]],c[p,j]);
        end;
        dfs(a[p][s[i,x]],x+1,i);
end;

procedure xuli;
var
        o,i,j,x:longint;
        blank:char;
begin
        for i:=1 to n do dfs(0,1,i);
        readln(q);
        for o:=1 to q do
        begin
                readln(x,blank,al);
                al:='`'+al;

                res:=0;
                for i:=1 to 26 do
                for j:=i+1 to 27 do
                res:=res+cnt[x,al[i],al[j]];

                writeln(res+1);
        end;
end;

begin
        //mofile;
        nhapdl;
        xuli;
        dongfile;
end.